

24

INBREEDING, OUTBREEDING, AND RANDOM MATING

Objectives

- Determine how nonrandom breeding affects allele and genotype frequencies in a population.
- Determine the effects of inbreeding on genotypic and phenotypic variation.
- Determine the effects of outbreeding on genotypic and phenotypic variation.
- Examine how assortative mating affects allele frequencies in a population.
- Explore inbreeding levels and the F statistic under various mating strategies.

Suggested Preliminary Exercise: Hardy-Weinberg Equilibrium

INTRODUCTION

One of the assumptions of the Hardy-Weinberg principle is that individuals in a population mate at random. In this exercise, you'll explore how violating this assumption affects the evolution of a population. **Random mating** occurs when individuals in the population pair off at random. That is, every individual has the same chance of breeding with any other individual in the population. **Inbreeding**, on the other hand, occurs when mated pairs are *more similar in genotypes* than if they were chosen at random. Because individuals of similar phenotypes will usually be somewhat similar in their genotypes, assortative mating (preferentially mating with an individual of similar phenotype) is generally thought to have the same consequences as inbreeding (Crow and Kimura 1970). **Outbreeding**, the flip side of inbreeding, occurs when mated pairs are *less similar in genotypes* than if they were chosen at random.

In this exercise, we will focus on how nonrandom mating affects the allele frequencies and genotype frequencies at a single locus. Keep in mind, however, that when organisms tend to mate nonrandomly, the entire genome is affected. Nonrandom breeding does one of two things: it either decreases the heterozygosity in the population (inbreeding) or it increases the heterozygosity of the population (outbreeding). You might think that nonrandom mating will also change the allele frequencies in the population. In fact, nonrandom mating without selection does not change the allele frequencies in a population at all. This will become apparent as you work through the exercise.

Because nonrandom mating affects heterozygosity levels, it is useful to “quantify” the level of nonrandom mating by comparing the heterozygosity observed in a population to the levels expected by Hardy-Weinberg. You might recall that if there are only two alleles, A_1 and A_2 , in the population at a given locus, the frequencies of the alleles are given by p and q , where p is the frequency of one kind of allele (A_1) and q is the frequency of the second kind of allele (A_2). For genes that have only two alleles,

$$p + q = 1 \quad \text{Equation 1}$$

For example, assume that the A locus has allele frequencies of $p = A_1 = 0.6$ and $q = A_2 = 0.4$. Given the allele frequencies for a population, the Hardy-Weinberg principle allows us to predict the genotype frequencies of a population, assuming that the population is large and that mating occurs at random, and that there is no gene flow, natural selection, or mutation acting on the population. The predicted genotypes of a population in Hardy-Weinberg equilibrium are $p^2:2pq:q^2$, where p^2 is the frequency of the A_1A_1 genotype, $2pq$ is the frequency of the A_1A_2 genotype, and q^2 is the frequency of the A_2A_2 genotype. The sum of the genotype frequencies, as always, will sum to 1. In this example, a population in Hardy-Weinberg equilibrium will have roughly the following genotype frequencies:

- Freq (A_1A_1) = $p^2 = p \times p = 0.6 \times 0.6 = 0.36$
- Freq (A_1A_2) = $2 \times p \times q = 2 \times 0.6 \times 0.4 = 0.48$
- Freq (A_2A_2) = $q^2 = 0.4 \times 0.4 = 0.16$.

Note that the genotype frequencies add to 1:

$$p^2 + 2pq + q^2 = 1 \quad \text{Equation 2}$$

Thus, approximately 48% of the individuals are expected to be heterozygous if the population is in Hardy-Weinberg equilibrium.

A population that mates nonrandomly will deviate from the Hardy-Weinberg expectation. This deviation is often quantified through the F statistic, also called the **inbreeding coefficient**:

$$F = \frac{H_0 - H}{H_0} \quad \text{Equation 3}$$

where H_0 is the heterozygosity level predicted by Hardy-Weinberg, and H is the observed level of heterozygosity. From an inbreeding perspective, the F statistic takes on values from 0 to 1. If the observed level of H is equal to H_0 , the numerator of Equation 3 is 0, and thus F is 0, indicating a randomly breeding population. When H is less than H_0 , there is a deficiency of heterozygotes in the population (due to inbreeding). Thus, positive F values indicate some level of inbreeding. The F statistic will be 1 (complete inbreeding) when the population consists of only homozygotes.

Let’s walk through an example. Suppose a population has the frequencies $A_1 = 0.6$ and $A_2 = 0.4$. As we calculated earlier, the expected frequency of heterozygotes is 0.48. Assume that this population, however, consists of 0 heterozygotes. The F statistic would be

$$F = \frac{0.48 - 0}{0.48} = 1$$

This population has the highest possible F statistic, suggesting that the population is highly inbred. If the population consisted instead of 48% heterozygotes, as predicted by Hardy-Weinberg, the F statistic would be

$$F = \frac{0.48 - 0.48}{0.48} = 0$$

Although the F statistic is intended to measure inbreeding, it also measures outbreeding as well, and takes on negative values when the observed level of heterozygosity is larger than that expected by Hardy-Weinberg. The F statistic can also be calculated through pedigree analysis (Hartl 2000). Inbreeding may affect an organism’s

fitness, or it may not. For example, average yield in hybrid corn decreases as F increases (Neal 1935), but low levels of heterozygosity in cheetahs (*Acinonyx jubatus*) do not appear to compromise their survival (Merola 1994).

When nonrandom mating occurs in a population, the Hardy-Weinberg genotype frequencies p^2 , $2pq$, and q^2 are not expected. However, if we know F , we can predict the frequencies of the A_1A_1 , A_1A_2 , and A_2A_2 genotypes (Hartl 2000). Let's start with the frequency of the A_1A_2 genotype. Remember that H is the observed genotype frequency of the heterozygotes in the population, so all we need to do is solve for H :

$$F = \frac{H_0 - H}{H_0}$$

Multiply both sides of the equation by H_0 to give

$$H_0 \times F = H_0 - H \tag{Equation 4}$$

Then subtract H_0 from both sides:

$$-H_0 + H_0 \times F = -H \tag{Equation 5}$$

Then multiply both sides of the equation by -1 and rewrite the equation so that H appears on the left side:

$$H = H_0 - H_0 \times F \tag{Equation 6}$$

And finally, since $2pq$ is the same thing as H_0 , or the heterozygosity expected under Hardy-Weinberg equilibrium, we can calculate H as a function of p , q , and F :

$$H = 2pq - 2pqF \tag{Equation 7}$$

Thus, if you know F , Equation 7 can predict the frequency of the A_1A_2 heterozygotes in a population. The frequency of the A_1A_1 and A_2A_2 homozygotes can also be predicted if you know F . Recall that the frequency of the A_1 allele (p) in a population is simply the frequency of the homozygotes (A_1A_1) plus half the frequency of the heterozygotes (A_1A_2). For simplicity, let's call the frequency of the A_1A_1 homozygotes D :

$$p = D + (H/2) \tag{Equation 8}$$

So now we need to solve for D , the frequency of the A_1A_1 homozygotes:

$$D = p - (H/2) \tag{Equation 9}$$

Since we know H from Equation 7, we can substitute in Equation 9 and simplify:

$$D = p - \frac{2pq - 2pqF}{2}$$

The 2 divides out, and subtracting $2pq$ and $2pqF$ from p gives us

$$D = p - pq + pqF \tag{Equation 11}$$

Now we can group the first two terms and factor out a p :

$$D = p(1 - q) + pqF \tag{Equation 12}$$

And finally, because $1 - q$ is the same thing as p , we arrive at

$$D = p^2 + pqF \tag{Equation 13}$$

The same logic will allow you to calculate the frequency of the A_2A_2 homozygotes, which we'll call R :

$$R = q^2 + pqF \tag{Equation 14}$$

Thus, Equations 7, 13, and 14 allow you to predict the genotype frequencies of a population where p , q , and F are known.

PROCEDURES

In this exercise, you will set up a spreadsheet model to explore the effects of inbreeding and outbreeding on a population. Your population will consist of 1000 individuals that select mates according to probabilities that you assign. We will consider the effects the inbreeding and outbreeding on the allele frequencies at a single locus. This locus has two alleles, A_1 and A_2 . The basic model will be fairly easy to construct, but the fun will start when you begin to change mating partners and see how mate selection and breeding system affect allele and genotype frequencies.

As always, save your work frequently to disk.

INSTRUCTIONS

A. Set up the population parameters

1. Open a new spreadsheet and set up column headings as shown in Figure 1.

ANNOTATION

	A	B	C
1	Inbreeding, Outbreeding, and Assortative Mating		
2			
3	Initial genotype frequencies		Tally
4			0
5	A1A1	0	
6	A1A2	1	
7	A2A2	0	
8			
9			
10			
11			
12			
13			
14			
15			
16			
17			
18			
19	Adult population		
20			
21	Individual	Random number	Adult genotype

Figure 1

2. Enter values shown in cells B5–B7.

We will start with a population whose genotype frequencies are given in cells B5–B7. Our population will consist solely of A_1A_2 heterozygotes since the frequency in cell B6 is 1.

3. Create a “running tally” in cells C4–C7

4. Set up a linear series from 1–1000 in cells A22–A1021.

5. In cells B22–B1021, generate random numbers between 0 and 1.

6. Use the **LOOKUP** function to assign genotypes to each of the 1000 individuals based on the frequencies you entered in cells B5–B7 and the tally of genotype frequencies in cells C4–C7. Save your work.

Enter **=B5** in cell C5.

Enter **=SUM(\$B\$5:B6)** in cell C6. Copy cell C6 into cell C7.

The running tally is necessary to assign genotypes to individuals in the population. It also will help you quickly verify that your genotype frequencies add to 1. Note that cell C7 must *always* equal 1. If it does not equal 1, it means that the frequencies entered in cells B5–B7 don’t add to 1 (adjust accordingly).

Enter **1** in cell A22.

Enter **= 1+A22** in cell A23. Copy this formula down to cell A1021.

You have now established a population of 1000 individuals. Save your work.

Enter **=RAND()**. When you press F9, the calculate key, the spreadsheet generates new random numbers.

In cell C22, enter the formula **=LOOKUP(B22,\$C\$4:\$C\$7,\$A\$5:\$A\$7)**. Copy your formula down to cell C1021.

The **LOOKUP** function looks up a value (the random number in cell B22) in a vector that you specify (cells **\$C\$4:\$C\$7**) and returns a genotype associated with that random number in the vector **\$A\$5:\$A\$7**. (Remember that a vector is a single row or column of values.) This function is handy for assigning genotypes to individuals because if **LOOKUP** can’t find the exact lookup value (the random number given in cell B22), it matches the largest value in the lookup vector (cells **\$C\$4:\$C\$7**) that is less than or equal to **lookup_value**. The result is that genotypes are assigned to individuals in approximately the proportions that you specified. Examine your first 10 genotypes. They should all be A_1A_2 if the **LOOKUP** function worked properly. To see how the function works, change cells B5 and B7 to 0.5, and set cell B6 to 0. (Remember that the final tally of genotype frequencies must equal 1 in cell C7.) Now examine the genotypes of your first 10 individuals. The genotypes should be either A_1A_1 or A_2A_2 . When you feel you have a handle on how this function works, return cells B5 and B7 to 0, and return cell B6 to 1.

B. Compute allele and genotype frequencies of the population.

1. Set up new column headings as shown in Figure 2:

2. In cells E9–G9, use the **COUNTIF** formula to count the number of A_1A_1 , A_1A_2 , and A_2A_2 genotypes in the population.

	D	E	F	G	H	I
1						
2						
3			Computed frequencies			
4			A1	A2	Total	F
5		Initial allele frequencies:				
6		F1 allele frequencies:				
7						
8		A1A1	A1A2	A2A2	Total	
9		Initial genotype numbers:				
10		Initial genotype frequencies:				
11		F1 genotype numbers:				
12		F1 genotype frequencies:				

Figure 2

Use the Paste Function key to guide you through the formulae. The **COUNTIF** formula counts the number of cells within a range that meet the given criteria. It has the syntax **COUNTIF(range,criteria)**, where **range** is the range of cells you want to examine, and **criteria** defines what you want to count.

- E9 =**COUNTIF(\$C\$22:\$C\$1021,E8)**
- F9 =**COUNTIF(\$C\$22:\$C\$1021,F8)**
- G9 =**COUNTIF(\$C\$22:\$C\$1021,G8)**

316 Exercise 24

3. In cell H9, use the **SUM** function to sum cells E9–G9.

4. In cells E10–G10, enter formulae to calculate genotype frequencies.

5. In cell H10, use the **SUM** function to sum the genotype frequencies.

6. In cells F5 and G5, enter formulae to calculate allele frequencies.

7. In cell H5, use the **SUM** function to sum the allele frequencies. Save your work.

C. Select mates.

1. Set up new column headings as shown in Figure 3.

2. Enter values shown in cells F15–H17.

Enter **=SUM(E9:G9)**. Your result should be 1000.

Remember that frequencies range from 0 to 1. To calculate the frequency of the A_1A_1 genotype in the population, write a formula that counts the number of A_1A_1 genotypes, divided by the total number of individuals in the population.

In cell E10 enter the formula **=E9/\$H\$9**.

In cell F10 enter the formula **=F9/\$H\$9**.

In cell G10 enter the formula **=G9/\$H\$9**.

In cell H10 enter the formula **=SUM(E10:G10)**.

The genotype frequencies calculated in cells C9–F9 should add to 1. If they don't, double-check your formulae.

In cell F5 enter the formula **=(E9*2+F9)/(2*H9)**.

In cell G5 enter the formula **=1-F5**.

Since our population consists of 1000 individuals, there are 2000 “gene copies” present. In order to compute frequencies we need to determine how many of those gene copies are A_1 and how many are A_2 . To calculate the frequency of the A_1 allele, we multiply the number of A_1A_1 homozygotes by 2 (because each individual carries two copies of this allele) and add to this number the number of heterozygotes (each heterozygote carries one copy of this allele). This sum is then divided by the total number of gene copies in the population ($2N$) to generate the frequency of the A_1 allele. Since there are only two alleles present, and since $p + q = 1$, we can obtain the frequency of the A_2 allele by subtraction.

Enter the formula **=SUM(F5:G5)**.

	D	E	F	G	H
14		Mate ==>	A1A1	A1A2	A2A2
15	Parental genotype:	A1A1	1	0	0
16	Parental genotype:	A1A2	0	1	0
17	Parental genotype:	A2A2	0	0	1
18					

Figure 3

Now we will let our population mate and produce offspring.

The parental genotypes are listed in cells E15–E17. The genotype of a potential mate is given in cells F14–H14.

Cells F15–H17 give the probabilities of mating with a particular genotype. These cells are shaded in Figure 3 to indicate that you directly enter values into these cells. For example, cell F15 gives the probability that an A_1A_1 genotype will mate with another A_1A_1 genotype. Cell G15 gives the probability that an A_1A_1 genotype will mate with a heterozygous genotype, and cell H15 gives the probability that an A_1A_1 genotype will mate with an A_2A_2 genotype. Note that *the sum of the probabilities across rows must equal 1*.

3. Set up column headings as shown in Figure 4.

	E	F	G	H	I	J
19	Mate selection					
20				Random	Preferred	Actual
21	Index 1	Index 2	Index 3	number	mate genotype	mate genotype

Figure 4

Our goal is to have the spreadsheet look up the genotype of individual 1 and match it to genotypes listed in cells E15–E17. Then we want to determine the genotype of individual 1’s mate, listed in cells F14–H14. To choose mates according to the probabilities given, we will use four different functions in: **MATCH**, **INDEX**, **RAND**, and **IF**. Used in combination, these formulae will allow us to generate the genotype of a mate for each individual in the population. Note that if a preferred genotype is not present in the population, the individual does not reproduce, and a rare genotype that is present and preferred can mate more than once.

4. In cell D22, enter a **MATCH** formula and copy the formula down to cell D1021.

In cell D22, enter the formula `=MATCH(C22,E15:E17)`.

The **MATCH** formula returns the *relative position* of an item in a table that matches the condition you specify. It has the syntax `MATCH(lookup_value,lookup_array)`. The formula in cell D22 tells the spreadsheet to find the genotype listed in cell C22, and return the *relative position* of that genotype in the table `E15:E17`. For example, the genotype of individual 1 in our program is A_1A_2 . The program returns the value 2, to indicate that A_1A_2 individuals occupy the second position in our array. If individual 1 had the genotype A_1A_1 , it would return the number 1, and if individual 1 had the genotype A_2A_2 , it would return the number 3. Copy this formula down for the remaining 999 individuals, and make sure your **MATCH** values are correct. Since your population consists solely of heterozygotes, the match values should all be equal to 2.

5. In cell E22, enter an **INDEX** formula and copy this formula down to cell E1021.

In cell E22, enter the formula `=INDEX(E15:H17,D22,2)`.

Our second trick is the **INDEX** formula. This formula returns the *value* of an element in a table, once you identify the row and column number that should be returned. The **INDEX** formula has the syntax `INDEX(array,row_num,column_num)`, where **array** is a range of cells in a table; **row_num** selects the row in the table from which to return a value, and **column_num** selects the column in table from which to return a value.

The formula in cell E22 tells the spreadsheet to examine the range of cells E15–H17, and to go to the row designated in cell D22 (derived from the **MATCH** formula entered in Step 4) and column 2 (which indicates the probability of mating with an A_1A_1 individual). The program will then return the value associated with this row and column intersection. Fill this formula down for the remaining 999 individuals in the population. Make sure you understand what is going on.

6. In cells F22 and G22, enter analogous **INDEX** formulae to generate the probability of mating with heterozygote and A_2A_2 homozygote, respectively. Copy your formulae down to cells F1021 and G1021.

In cell F22 enter the formula `=INDEX(E15:H17,D22,3)`.

In cell G22 enter the formula `=INDEX(E15:H17,D22,4)`.

The three **INDEX** formulae generate the appropriate mating probabilities for each individual in the population. Figure 5 shows the genotypes of the first four individuals in our population, their match values, and index values.

	C	D	E	F	G	H
14			Mate ==>	A1A1	A1A2	A2A2
15	Parental genotype:	A1A1		1	0	0
16	Parental genotype:	A1A2		0	1	0
17	Parental genotype:	A2A2		0	0	1
18						
19	Mate selection					
20						Random
21	Adult genotype	Match	Index 1	Index 2	Index 3	number
22	A1A2	2	0	1	0	
23	A1A2	2	0	1	0	
24	A1A2	2	0	1	0	
25	A1A2	2	0	1	0	

Figure 5

7. In cell H22, generate a random number between 0 and 1. Copy your formula down to H1020.

8. In cell I22, enter a nested IF formula to select the genotype of the preferred mate. Copy your formula down to cell I1021.

9. In cell J22, enter an IF formula to generate the actual mate genotype, and copy this formula down to cell J1021.

In cell H22 enter the formula =RAND().

The RAND function generates a random number between 0 and 1. We will use this formula to determine the genotype of the mate for each individual in the population. When you press F9, the calculate key, the spreadsheet generates a new set of random numbers.

In cell I22 enter the formula =IF(H22<=E22,\$F\$14,IF(H22<=E22+F22,\$G\$14,\$H\$14)). Copy the formula down cell I1021 to obtain preferred mates for the remaining individuals in the population.

Our final step in selecting the genotype of the mate is to use two nested IF functions. Remember that an IF statement returns one value if a condition you specify is true and another value if the condition you specify is false. IF statements have the form IF(logical_test,value_if_true,value_if_false).

The first portion of the IF formula in I22, =IF(H22<=E22,\$F\$14, tells the spreadsheet to examine cell H22 (the random number associated with individual 1). If that value is less than or equal to the value in cell E22 (the first index number), return the value in cell F14 (A₁A₁). Otherwise, go through the second IF statement, IF(H22<=E22+F22,\$G\$14,\$H\$14). This statement tells the program to examine the random number in cell H22. If that value is less than the sum of the values in cells E22 and F22 (the first and second index numbers), return the value in cell G14 (A₁A₂). Otherwise, return the value in cell H14 (A₂A₂).

Enter the formula =IF(VLOOKUP(I22,\$A\$5:\$B\$7,2)>0,I22,""). Although we have established the mating preferences, we now need to ensure that an individual with the preferred genotype actually exists in the population for mating. The formula in cell J22 is a VLOOKUP function nested within an IF function. It tells the spreadsheet to look up the preferred mate's genotype given in cell I22 in the table of cells A5–B7 and return the associated value in the second column, which is the frequency of the preferred genotype. If the frequency of the preferred genotype is greater than 0, preferred individuals exist in the population for mating, and the spreadsheet returns the genotype listed in cell I22. If the preferred genotype does not exist in the population, its genotype frequency is 0, so the formula returns a period to indicate that the individual will not mate.

Take some time to make sure you can see how the formulae in cells I22 and J22 are working. In the example shown in Figure 6, individual 1 (A₁A₂) prefers to mate with an A₁A₂ genotype because its random number is greater than Index 1 (0) and less than the sum of Index 1 and Index 2 (which is 1). Since the preferred genotype is present in the population (its frequency is greater than 0), the actual mate genotype (A₁A₂) is given in cell J22.

10. Save your work.

D. Obtain genotypes of offspring.

1. Set up column headings as shown in Figure 7.

2. In cell K22, enter a formula that will produce an "offspring" by randomly combining a genotype from each of the two parents. Copy the formula down to cell K1021.

3. Enter a formula so all heterozygotes will be listed as A_1A_2 . Copy the formula down to L1021.

4. Save your work.

E. Calculate the new genotype and allele frequencies and the F statistic.

1. In cells E11–G11, use the **COUNTIF** formula to count the number of genotypes in the offspring population.

2. In cell H11, sum the offspring genotypes.

	D	E	F	G	H	I	J
19	Mate Selection						
20					Random	Preferred	Actual
21	Match	Index 1	Index 2	Index 3	number	mate genotype	mate genotype
22	2	0	1	0	0.905419	A1A2	A1A2
23	2	0	1	0	0.079056	A1A2	A1A2
24	2	0	1	0	0.793008	A1A2	A1A2
25	2	0	1	0	0.85476	A1A2	A1A2

Figure 6

	K	L
19	Offspring	
20		
21	Genotype	Genotype

Figure 7

In cell K22 enter the formula `=IF(RAND()<0.5,LEFT(C22,2),RIGHT(C22,2))&IF(RAND()<0.5,LEFT(J22,2),RIGHT(J22,2))`.

You are already familiar with the **RAND** function. The **LEFT** and **RIGHT** functions return either the *leftmost* or *rightmost* characters in a string of characters. For example, **LEFT(C22,2)** returns the leftmost two characters listed in cell C22.

The formula in cell K22 draws a random number for individual 1; if the random number is less than or equal to 0.5, individual 1 contributes the "left" allele in its genotype as a gamete; otherwise, it contributes the "right" allele in its genotype as a gamete. A second **IF** statement is used to determine the gamete contributed by individual 1's mate in column J. The gamete from individual 1 and its mate are joined with an **&** symbol, which produces the genotype of the offspring.

In cell L22, enter the formula `=IF(K22="A2A1","A1A2",K22)`.

This formula is necessary because some of the heterozygous offspring will be listed as A_1A_2 and some will be listed as A_2A_1 . For simplicity, we will make all the heterozygotes be listed as A_1A_2 .

In cell E11 enter the formula `=COUNTIF(L22:L1021,E8)`.

In cell F11 enter the formula `=COUNTIF(L22:L1021,F8)`.

In cell G11 enter the formula `=COUNTIF(L22:L1021,G8)`.

Enter the formula `=SUM(E11:G11)`. Double-check your formulae. Your results should total to 1000.

320 Exercise 24

3. In cells E12–G12, calculate genotype frequencies of the offspring population.
4. In cell H12, sum the offspring genotype frequencies.
5. In cell F6, enter a formula to calculate the frequency of the A_1 allele in the offspring population.
6. In cell G6, calculate the frequency of the A_2 allele.
7. In cell I5, calculate F , the inbreeding coefficient of the offspring. Save your work.

F. Create graphs.

1. Graph the genotype frequencies of the parental population and the offspring population.

2. Graph the allele frequencies of the parental population and the offspring population for the A_1 allele.

3. Save your work.

In cell E12 enter the formula =E11/\$H\$11.
 In cell F12 enter the formula =F11/\$H\$11.
 In cell G12 enter the formula =G11/\$H\$11.

Enter the formula =SUM(E12:G12). Your results should total to 1.

In cell F6 we entered the formula =(E11*2+F11)/(2*H11)

Remember that in a population of 1000 individuals, there are 2000 “gene copies” present because each individual carries two alleles. We just need to know how many of those gene copies are A_1 and how many are A_2 .

In cell G6 enter the formula =1-F6.

Since there are only two alleles at the A locus, $p + q = 1$. Since you already calculated p, q can be obtained by subtraction.

Now we are ready to calculate the inbreeding coefficient of our offspring population. Remember that

$$F = \frac{H_0 - H}{H_0}$$

where H_0 is the heterozygosity level predicted by Hardy-Weinberg, and H is the observed level of heterozygosity. We used the formula $=((2*F6*G6)-F12)/(2*F6*G6)$. Your result should be close to 0, since the offspring population will consist of approximately 25% A_1A_1 , 50% A_1A_2 , and 25% A_2A_2 genotypes, as predicted by Hardy-Weinberg. Take a moment to consider your results.

Use a column graph. Select cells E10–G10 to graph the parental genotypes and cells E12–G12 to graph the offspring genotypes. Your graph should resemble Figure 8.




Figure 8

Use a line graph and select cells F5 and F6. Label your graph fully. Your graph should resemble Figure 9. Set the scale of the y -axis to range between 0 and 1.




Figure 9

QUESTIONS

1. How does the allele frequency change from the parental population to the offspring population? How does the genotype frequency change from the parental population to the offspring population? Change the parental genotype frequencies in cells B5–B7 to 0.33, 0.34, and 0.33. How did the allele frequency change from the parental population to the offspring population? How did the genotype frequency change from the parental population to the offspring population?
2. Press F9, the calculate key, to generate new results. Why do your results vary from trial to trial?
3. Assume your offspring population will now breed and produce the next generation. How do F , p , and the genotype frequencies change over time with complete inbreeding? Set up new column headings as shown.

	J	K	L	M	N	O
3		Generation				
4		1	2	3	4	5
5	A1A1	0	0.247			
6	A1A2	1	0.498			
7	A2A2	0	0.255			
8	p	0.5	0.496			
9	F		0.003936			

Enter the genotype frequencies of your parental population in cells K5–K7. Enter the frequency of the A_1 allele, p , in cell K8. Enter your genotype frequencies of the offspring population in cells L5–L7. Enter p and F for the offspring population in cells L8–L9. (Your values will likely be a bit different than shown. If you copy and paste your results into the cells, make sure you choose paste special | paste values). Now let the offspring genotypes be the parental genotypes. Enter the offspring genotype values in cells B5–B7. Record the genotype

322 Exercise 24

frequencies of the new offspring population in cells M5–M7. Repeat until 5 generations have elapsed. How do F , p , and the genotype frequencies change over time with complete inbreeding? Graph your results.

4. Set the initial genotype frequencies to 0.25, 0.5, and 0.25 (cells B5–B7). What is the relationship between the probability of mating with the same genotype and F ? Set up new column headings as shown, where Probability is the probability of mating with the same genotype:

	K	L	M
10		Offspring	
11	Probability	p	F
12	0		
13	0.2		
14	0.4		
15	0.6		
16	0.8		
17	1	0.5	0.5

You have already examined the case where $p = 1$. Enter the offspring p and F values in cells L17–M17. Now change the mating probabilities in cells F15–H17. Start with strict outbreeding, where the probability is 0. Enter 0 in cells F15, G16, and H17. Set the other mating probabilities so that the probability of mating with a dissimilar genotype is the same for the two remaining alternative genotypes (e.g., for Probability = 0, set the probability of mating with the other two kinds of genotypes to 0.5 so that they have equal changes of being selected for mating). For example, when the probability of mating with a similar genotype is 0.4, your spreadsheet should look like this:

	D	E	F	G	H
14		Mate ==>	A1A1	A1A2	A2A2
15	Parental genotype:	A1A1	0.4	0.3	0.3
16	Parental genotype:	A1A2	0.3	0.4	0.3
17	Parental genotype:	A2A2	0.3	0.3	0.4

Record p (the frequency of the A_1 allele in the offspring population) and F in cells L12 and M12. Repeat the process for the remaining probabilities. Graph the relationship F and the probability of mating with the same genotype. Graph the relationship between the frequency of the A_1 allele in the offspring population and the probability of mating with the same genotype. Interpret your results.

5. Assume that A_1 is dominant to A_2 , and that individuals breed with the same phenotype. Set the mating probabilities in cells F15–H17 accordingly (e.g., A_1A_1 individuals are equally likely to mate with A_1A_1 or A_1A_2 individuals, but are not likely to mate with A_2A_2 individuals). How does assortative mating differ from inbreeding effects on genotype and allele frequencies of the offspring population? How does it differ from random mating? To simulate random mating, enter the parental genotype frequencies in the cells; since mates are drawn at random, an individual should encounter a random mate proportionally to the parental frequencies.

LITERATURE CITED

Crow, J. F. and M. Kimura. 1970. *An Introduction to Population Genetics Theory*. Harper & Row, New York.

Hartl, D. 2000. *A Primer of Population Genetics*, 3rd Edition. Sinauer Associates, Sunderland, MA.

Merola, M. 1994. A reassessment of homozygosity and the case for inbreeding depression in the cheetah, *Acinonyx jubatus*: Implications for conservation. *Conservation Biology* 8: 961–971.

Neal, N. P. 1935. The decrease in yielding capacity in advanced generations of hybrid corn. *Journal of the American Society of Agronomy* 27: 666–670.

